

```
// GLOSS.WCM — expand glossary entries, the right way.
//
// The GLOSSARY.WCM supplied with WPwin requires the user to edit the
// macro and add a subroutine
// for each new glossary entry. This macro, by contrast, stores the glossary in
// a much simpler format,
// in the WPWP.INI file (which is kept in the Windows directory).
//
// The companion macro GLOSSADD.WCM allows easy addition of glossary
// entries.
//
// This code is based in part on GLOSSARY.WCM, BULLET.WCM, and
// BULLETD.F.WCM from the
// WPWin distribution. Note that the INI file reading and writing techniques
// appear to be undocumented,
// and will not work except in WPWin releases of 30 April 92 and later.
//
// Author: Richard Reiner <rreiner@nexus.yorku.ca>
//
// Revision history
// 28 July 1992          RjR   Initial version
//
```

```
APPLICATION(WP;WPWP;Default)
```

```
// *****
// Main
//
CALL(Init@)
CALL(GetKey@)
CALL(LookUpKey@)
CALL(ReplaceKey@)
CALL(End@)
// NOTREACHED
QUIT
```

```
// *****
// Init@
//
// Initialization
//
LABEL(Init@)
ONERROR(End@)
ONCANCEL(End@)
MacroStatusPrompt(Off!)
```

```

Display(Off!)
DEBUG:=0
ERRMARKER:="**ERR**"
DLLLoad(MCRLink; "WPCMCR.DLL")
    RETURN

// *****
// GetKey@
//
// Get the key (unexpanded glossary entry) from the current doc, at the
// caret.
//
    LABEL(GetKey@)
SelectWordPrevious()
GetWPData(Key; SelectedText!)
    RETURN

// *****
// LookUpKey@
//
// Get the key (unexpanded glossary entry) from the current doc, at the
// caret.
//
    LABEL(LookUpKey@)
// Initialize parm as string
KeyExp := ""
IniResult := 0
DLLCall(MCRLink;
        "MCRGetIniWPString";
        IniResult : WORD;
        {
            "WPWP"; // Name of INI file
            AnsiString("RJR-Glossary"); // section
            AnsiString(Key); // key
            AnsiString(ERRMARKER); // error return
            Address(KeyExp) // parm passed by reference
        }
)
IF(DEBUG=1)
    PROMPT("GLOSS.WCM";"K: "+Key+" E: "+KeyExp; 4;;)
    PauseKey(Enter!)
    ENDPROMPT
ENDIF
    RETURN

```

```

// *****
// ReplaceKey@
//
// Get the key (unexpanded glossary entry) from the current doc, at the
// caret.
//
LABEL(ReplaceKey@)
IF(KeyExp = ERRMARKER)
    PROMPT("GLOSS.WCM: lookup failure";"There is no expansion defined for
    """+Key+""";1;;)
    PauseKey(Enter!)
    ENDPROMPT
    // Just turn select off and put the caret back where it was
    Selectmode(Off!)
    PosGoto(LastPosition!)
ELSE
    // Select is on, so the expansion replaces the key
    Type(KeyExp)
ENDIF
RETURN

// *****
// End@
//
// Wrap things up and terminate
//
LABEL(End@)
DLLFree(MCRLink)
QUIT

```